# Maintainer: David Runge <dave@sleepmap.de>
# Contributor: speps <speps at aur dot archlinux dot org>

pkgname=g2reverb
pkgver=0.7.1
pkgrel=6
pkgdesc="LADSPA stereo reverb plugin based on greverb"
arch=('loong64' 'x86_64')
url="https://kokkinizita.linuxaudio.org/linuxaudio/ladspa/"
license=('GPL2')
groups=('ladspa-plugins' 'pro-audio')
depends=('gcc-libs')
makedepends=('ladspa')
source=("https://kokkinizita.linuxaudio.org/linuxaudio/downloads/${pkgname}-${pkgver}.tar.bz2")
sha512sums=('ff6f8de5f41e031e0841acb0e60a0ee9c7be9b7c4375ccda4c0f406b4e789ab84e2b2d52c31ec1bf2be75309a2068d5a167c94568346c7251df9c513b4c6c189')

build() {
  cd "${pkgname}-${pkgver}"
  make
}

package() {
  cd "${pkgname}-${pkgver}"

  # libs
  install -vDm 755 "${pkgname}.so" "${pkgdir}/usr/lib/ladspa/${pkgname}.so"
  # docs
  install -vDm 644 {AUTHORS,README} -t "${pkgdir}/usr/share/doc/${pkgname}/"
}
